

# DRAGON OPEN FILE

Send us your Dragon programs, beginning with a general description and then explaining how the program is constructed. Take care that the listings are all bug free, enclosing a cassette and, if possible, a printout. We pay £6 for each original bug-free program published, double for the program of the month. If you have any problems with the listings, please send your queries to the appropriate author, *Dragon User*, 12-13 Little Newport Street, London WC2H 7PP.

## Meander

From Keith David in Sussex

MEANDER is a strategy board game for two players. The object of the game is to form a continuous line from the top left corner of the board to the bottom right

corner, or to force your opponent off the edge of the board.

To form the line, one of three possible pieces are placed at the end of the existing line. These pieces will cause the line to change direction, and sometimes to loop over itself. The program will automatically search for the end of the line, so in order to place a piece, it is only necessary to type in its identity letter. The three pieces are permanently displayed on the side of the board, for reference.

### Program Notes

Lines

150-450 Draw the board.

630-770 Search for the end of the line. Line 770 checks whether the next space is vacant, and if it is not, causes a repeat search.

490 and 760 Check for the two different ways of ending the game.

When typing the program, all REMS may be omitted.




Fig. 1. The board at the start of a game.

```

10 REM MEANDER
20 REM INITIALIZATION AND TITLES
30 DIMX(0,11),Y(0,11),Z(0,11),A(9,9)
40 P1$="BM229,122;R2NR2U14G2"
50 P2$="BM227,122;NR8U5E2R4E2U3H2L4G2"
60 CLS
70 PRINT@73,STRING$(11,"*")
80 PRINT@105,"* MEANDER *"
90 PRINT@137,STRING$(11,"*")
100 FOR I=1 TO 1000:NEXT I
110 PRINT@224,"DO YOU WANT INSTRUCTIONS? (Y/N)"
120 I$=INKEY$:IF I$="" THEN 120
130 IF I$="Y" THEN 890
140 REM DRAW BOARD

```

```

150 PMODE4,1:PCLS:SCREEN1,1
160 FOR Y=12 TO 112 STEP 50
170 LINE(8,Y)-(28,Y+20),PSET,B
180 NEXT Y
190 DRAW"BM18,12;D1ONL10NR10ND10"
200 CIRCLE(8,62),10,5,1,0,.25
210 CIRCLE(28,82),10,5,1,.5,.75
220 CIRCLE(28,112),10,5,1,.25,.5
230 CIRCLE(8,132),10,5,1,.75,1
240 DRAW"BM14,36;M22,48;BM14,48;M22,36"
250 DRAW"BM14,86;M18,92;ND6;M22,86"
260 DRAW"BM14,136;R8;M14,148;R8"
270 GET(8,12)-(28,32),X,G
280 GET(8,62)-(28,82),Y,G
290 GET(8,112)-(28,132),Z,G
300 FOR X=48 TO 208 STEP 20
310 LINE(X,12)-(X,172),PSET
320 NEXT X
330 LINE(46,10)-(210,174),PSET,B
340 FOR Y=12 TO 172 STEP 20
350 LINE(48,Y)-(208,Y),PSET
360 NEXT Y
370 PUT(48,12)-(68,32),X,PSET
380 PUT(188,152)-(208,172),X,PSET
390 DRAW"BM228,22;U10R6D5L6"
400 DRAW"BM228,36;NU10R6"
410 DRAW"BM228,50;U5NR6U4E1R4F1D9"
420 DRAW"BM231,64;U5NM228,54;M234,54"
430 DRAW"BM228,78;NR6U5NR4U5R6"
440 DRAW"BM228,92;U10R5F1D3G1NL5M234,92"
450 LINE(221,105)-(241,125),PSET,B

```

```

460 REM PLAY
470 FOR X=0 TO 9:FOR Y=0 TO 9:A(X,
Y)=0:NEXT Y,X
480 A(1,1)=1:A(8,8)=1:X=58:Y=22:X1
=78:Y1=22:P=2
490 IF X1>208 OR X1<48 OR Y1>172 O
R Y1<12 THEN 820
500 ON P GOTO 510,520
510 DRAW"C0"+P1$+"C5"+P2$:P=2:GOTO
530
520 DRAW"C0"+P2$+"C5"+P1$:P=1
530 I$=INKEY$:IF I$="" THEN 530
540 IF I$<"X" OR I$>"Z" THEN 530
550 ON ASC(I$)-87 GOTO 560,580,600
560 PUT(X1-10,Y1-10)-(X1+10,Y1+10)
,X,PSET:A((Y1-2)/20,(X1-38)/20)=1
570 GOTO 630
580 PUT(X1-10,Y1-10)-(X1+10,Y1+10)
,Y,PSET:A((Y1-2)/20,(X1-38)/20)=2
590 GOTO 630
600 PUT(X1-10,Y1-10)-(X1+10,Y1+10)
,Z,PSET:A((Y1-2)/20,(X1-38)/20)=3
610 GOTO 630
620 REM SEARCH FOR END OF LINE
630 ON A((Y1-2)/20,(X1-38)/20) GO
TO 640,680,720
640 IF X1=X+20 THEN X=X1:Y=Y1:X1=X
+20:Y1=Y:GOTO 760
650 IF X1=X-20 THEN X=X1:Y=Y1:X1=X
-20:Y1=Y:GOTO 760
660 IF Y1=Y-20 THEN X=X1:Y=Y1:X1=X
:Y1=Y-20:GOTO 760
670 IF Y1=Y+20 THEN X=X1:Y=Y1:X1=X
:Y1=Y+20:GOTO 760
680 IF X1=X+20 THEN X=X1:Y=Y1:X1=X
:Y1=Y-20:GOTO 760
690 IF X1=X-20 THEN X=X1:Y=Y1:X1=X
:Y1=Y+20:GOTO 760
700 IF Y1=Y-20 THEN X=X1:Y=Y1:X1=X
+20:Y1=Y:GOTO 760
710 IF Y1=Y+20 THEN X=X1:Y=Y1:X1=X
-20:Y1=Y:GOTO 760
720 IF X1=X+20 THEN X=X1:Y=Y1:X1=X
:Y1=Y+20:GOTO 760
730 IF X1=X-20 THEN X=X1:Y=Y1:X1=X
:Y1=Y-20:GOTO 760
740 IF Y1=Y-20 THEN X=X1:Y=Y1:X1=X
-20:Y1=Y:GOTO 760
750 IF Y1=Y+20 THEN X=X1:Y=Y1:X1=X
+20:Y1=Y:GOTO 760
760 IF X1=198 AND Y1=162 THEN 790
770 IF A((Y1-2)/20,(X1-38)/20)=0 T
HEN 490 ELSE 630
780 REM END TITLES
790 FOR I=1 TO 20:SCREEN 1,0:SOUND
100,1:SCREEN 1,1:SOUND 150,1:NEXT
I
800 CLS:IF P=1 THEN PRINT@71,"PLAY
ER 1 WINS." ELSE PRINT@71,"PLAYER
2 WINS."
810 GOTO840

```

```

820 FOR I=1 TO 20:SCREEN 1,0:SOUND
100,1:SCREEN 1,1:SOUND 150,1:NEXT
I
830 CLS:IF P=1 THEN PRINT@64,"PLA
YER 1 LOSES BY GOING OFF EDGE" ELS
E PRINT@64,"PLAYER 2 LOSES BY GOIN
G OFF EDGE"
840 FOR I=1 TO 1000:NEXT I
850 PRINT@448,"PRESS SPACEBAR FOR
ANOTHER GAME."
860 I$=INKEY$:IF I$="" THEN 860
870 IF I$="" THEN 150 ELSE STOP
880 REM INSTRUCTIONS
890 CLS
900 PRINT@32," THE OBJECT OF 'MEAN
DER' IS TO COMPLETE A CONTINUOUS
LINE FROM THE TOP LEFT CORNER OF T
HE BOARD TO THE BOTTOM RIGHT CORNER
, OR TO FORCE YOUR OPPONENT TO CA
RRY THE LINE OFF THE EDGE OF THE
BOARD."
910 PRINT:PRINT" SIMPLY TYPE IN TH
E LETTER OF THE PIECE YOU WISH T
O PLAY TO CONTINUE THE LINE."
920 PRINT@451,"PRESS ANY KEY TO CO
NTINUE."
930 I$=INKEY$:IF I$="" THEN 930
940 CLS
950 PRINT:PRINT" THE FIRST PIECE W
ILL BE PLACED ON THE SECOND SQUARE
OF THE TOP LINE, SO PLACING PIECE
'Y' AS THE FIRST MOVE OF A GAME
WILL AUTOMATICALLY LOSE THE GAM
E."
960 PRINT@356," PRESS ANY KEY TO P
LAY."
970 I$=INKEY$:IF I$="" THEN 970
980 GOTO 150

```




Fig. 2. A possible position after 29 moves. If Player 2 now plays piece 'X' he will lose. If he plays piece 'Y' he is bound to win.